

## A Macro to Easily Generate an interactive Google Map Report


Ting Sa, Cincinnati Children's Hospital Medical Center, Cincinnati, OH

### ABSTRACT


In this paper, a SAS macro is introduced to generate an interactive Google Map report. Using this macro, you can mark, or "drop a pin" on any location on a Google map by longitude and latitude coordinates. This paper also provides several ways you can find the longitude and latitude coordinates. Once a location is marked on the map, users can further use the mouse to hover over, or click the "pin" icon to display a pop out dialogue box with additional information about that location. The information shown in the dialogue box can be any multimedia information, such as plain text, images, videos, URL, email links, etc. The report generated by this macro retains all the functionalities of the Google map, allowing you to zoom in, zoom out, or move the map in the report, show the map in satellite mode, etc. The macro also has capability for you to display different styles of pin icons on the map. The SAS user does not need prior knowledge or expertise in any website programming language to use this macro. The SAS user only needs to prepare the input data, call the macro, and the Google map report will be generated.

### INTRODUCTION


Geographic information have become abundant with today's GPS navigation devices, smart phones with geolocation applications, and Google maps saved as a "favorite" on our computer web browser. Using a map report to display geographic locations is very useful. Such a capability can be useful in areas such as epidemiology and public health, and disease surveillance. John Snow, now known as "the father of field epidemiology," investigated an outbreak of cholera in the Golden Square of London in 1854. In mapping the location of cholera cases in the city in relation to the location of water pumps, he was able to determine the source of the outbreak of disease. With today's technology, we can generate a map report based on the Google map. In this paper, a macro to generate an interactive Google map report is presented. The following screenshots show examples of a Google map report.




**Figure 1. Using Location Pins to Indicate Places on the Google Map, the Pins can be Different Colors**



**Figure 2. Zoom in the Google Map Report and Display the Map Using the Satellite Mode**



**Figure 3. The Tooltip Displays When You Point Your Mouse to the Location Pin**



**Figure 4. Click the Location Pin to Display a Dialogue Box with Information of the Place. The Information can be Plain Text Messages, Website Links, Images, Email Links or Video Links.**

To generate a report like the above examples, the SAS user needs to prepare an input data set such as the one in Figure 5. The variable name "icon" specifies the color of the location pin. The latitude and longitude coordinates are shown in their respective columns. The tooltip column contains the text to be displayed in the tooltip. The "info" variable provides additional information to display in a pop out dialog box. Note that the "icon", "tooltip" and "info" columns are optional, and thus can be omitted. Also, the variable names can be of your own choosing.

icon	latitude	longitude	tooltip	info
<a href="http://maps.google.com/mapfiles/ms/icons/blue-dot.png">http://maps.google.com/mapfiles/ms/icons/blue-dot.png</a>	39.140634	-84.501069	Cincinnati Children's Hospital	Cincinnati Children's Hospital 3333 Burnet Ave Cincinnati OH 45229 <a href="http://www.cchmc.org">http://www.cchmc.org</a> <a href="#">Website</a>  <a href="mailto:abc@cchmc.org">E-mail</a> <a href="http://www.youtube.com/embed/Bw3ovKx5J5I">http://www.youtube.com/embed/Bw3ovKx5J5I</a>
<a href="http://maps.google.com/mapfiles/ms/icons/red-dot.png">http://maps.google.com/mapfiles/ms/icons/red-dot.png</a>	39.132484	-84.435285	Queen City Physicians TriHealth Physicia	Queen City Physicians TriHealth Physicia 2753 Erie Ave Cincinnati OH 45208
<a href="http://maps.google.com/mapfiles/ms/icons/purple-dot.png">http://maps.google.com/mapfiles/ms/icons/purple-dot.png</a>	39.146959	-84.51875	Good Samaritan Hospital	Good Samaritan Hospital 375 Dixmyth Ave Cincinnati OH 45220
<a href="http://maps.google.com/mapfiles/ms/icons/yellow-dot.png">http://maps.google.com/mapfiles/ms/icons/yellow-dot.png</a>	39.1261	-84.517318	The Christ Hospital	The Christ Hospital 2139 Auburn Ave Cincinnati OH 45219
<a href="http://maps.google.com/mapfiles/ms/icons/green-dot.png">http://maps.google.com/mapfiles/ms/icons/green-dot.png</a>	39.110911	-84.613884	Mercy Health	Mercy Health 3131 Queen City Ave Cincinnati OH 45238

**Figure 5. A Sample Input Data Set for the Macro**

There are some papers that show how to create a Google map report [1][2][3]. There may be a couple of challenges you encounter when trying to follow these papers. First, you need to modify the code presented in these papers to fit your own data set. If you are not familiar with website programming languages like html or javascript, this may be an obstacle. Second, you may need to have other products installed for the code to work. The macro presented in this paper aims to simplify the work of building an interesting interactive Google map report. You need only prepare the input data set, and call the macro.


For the input data set, you may not have the longitude and latitude coordinates. However, there are several ways of finding those data.

## OBTAINING THE LONGITUDE AND LATITUDE INFORMATION

1. Using PROC GEOCODE: If you have the street address of the location, you can use PROC GEOCODE to find the latitude and longitude coordinates. The address of the location doesn't need to be complete, it can be just a

ZIP code, a city, or a state. Of course, the more detailed the address, the more accurate the latitude and longitude values will be. To know how to use the PROC GEOCODE, check references [4][5].

2. Using Google map: You can type the location name or the address into Google map. Then, on Google map, right click the place icon and choose the option "What's here" from the pop up menu. Google map will display the longitude and latitude of that location for you. See the screenshots in Figure 6:



**Figure 6. Using Google Map to Find the Longitude and the Latitude**

3. Using Google map service: You can also write a SAS program to call Google map web service to find the longitude and latitude coordinates. Google map web service can return the results in xml format or json format. You need to have some knowledge of how to parse the xml data or the json data. Check references [6][7] for more details.

## THE GOOGLE\_MAP\_REPORT MACRO

### THE MACRO PARAMETERS

You can find the entire code of the macro at the end of the paper. However, immediately below describes the different parameters of the macro:

```
%macro generate_Google_map(indata=,reportnm=,lat=,lng=,tooltip=%str(),  
info=%str(),icon=%str());
```

- The "indata" is used to indicate the input SAS dataset name.
- The "reportnm" is used to indicate the location to save the output html report.
- The "lat" is used to indicate the variable name that saves the latitude data.
- The "lng" is used to indicate the variable name that saves the longitude data.
- The "tooltip" is used to indicate the variable name that saves the tooltip message.
- The "info" is used to indicate the variable name that saves the dialogue box message.
- The "icon" is used to indicate the variable name that saves the location pin icon style.

The tooltip, info and icon are optional, you don't need to specify the values to the macro if you don't have that information.

### CALL THE MACRO

Below is an example to show how to call the macro to generate a Google map report that has tooltip, dialogue box and different styles of location pin icons:

```
%generate_Google_map(indata=geodata,reportnm=%str('c:\mapreport.html'),lat=latitude,lng=longitude,tooltip=tooltip,info=info,icon=icon);
```

To generate a Google map report with tooltip, call the macro in this way, by default, the location pin icon will be red:

```
%generate_Google_map(indata=geodata,reportnm=%str('c:\mapreport.html'),lat=latitude,lng=longitude,tooltip=tooltip);
```

To generate a Google map report without tooltip and dialogue box, call the macro in this way and you will only see the places on the map report:

```
%generate_Google_map(indata=geodata,reportnm=%str('c:\mapreport.html'),lat=latitude,lng=longitude);
```

## CREATE A MULTIMEDIA DIALOGUE BOX

To add multimedia data to the dialogue box, such as bolded text, break texts, images, videos, website link, email links, simply prepare your input data set by adding some html tags around the texts in the information column. You don't need to write your own website programming codes!

To make bold text, you just add "<b>" in front of your text, and add "</b>" at the end of the text. For example, <b>Hello</b> will make "Hello" in bold text. See the screenshot in Figure 7:




Figure 7. Make Bold Text

To break a text string, you just add <br> at the end of your text. For example, Hello<br> World will make "World" show in the next line after "Hello." See the screenshot in Figure 8:




Figure 8. Break Text

To add an image to the dialogue box, for example, if you have an image called "hello.jpg" saved on your C: drive, just write  into the information column of your input data set. See the screenshot in Figure 9:




Figure 9. Add Image

You can add a YouTube video to the dialogue box. For example, if a YouTube video web address is <https://www.youtube.com/watch?v=8HzBQucKwTQ>, you can write <iframe src="http://www.youtube.com/embed/8HzBQucKwTQ" allowfullscreen></iframe> into the information column where "8HzBQucKwTQ" is the video code. See the screenshot in Figure 10:



Figure 10. Add a Video

To add an email link to the dialogue box, suppose the email address is abc@cchmc.org, you just write <a href="mailto:abc@cchmc.org">E-mail</a> into the information column of your input data set. See the screenshot in Figure 11:

info
<code>&lt;a href='mailto:abc@cchmc.org'&gt;E-mail&lt;/a&gt;</code>

**Figure 11. Add an Email Link**

To add a web link to the dialogue box, suppose you want to add the Google website, you just write `<a href=http://www.Google.com>Google</a>` into the information column of your input data set. See the screenshot below:

info
<code>&lt;a href=http://www.Google.com&gt;Google&lt;/a&gt;</code>

**Figure 12. Add a Web Link**

Generally, to add any number of multimedia messages, you just use the above examples to write SAS code to prepare each message and combined them to put in your information column. Figure 13 is the screenshot for an information column that have multiple multimedia messages:

info
<code>Cincinnati Childrens Hospital&lt;br&gt;3333 Bumet Ave&lt;br&gt;Cincinnati&lt;br&gt;OH 45229&lt;br&gt;&lt;a href=&amp;#39;http://www.cchmc.org&amp;#39;&gt;Website&lt;/a&gt;&lt;br&gt;&lt;img alt="STBuilding.jpg" src="STBuilding.jpg" /&gt;&lt;br&gt;&lt;a href="mailto:abc@cchmc.org"&gt;E-mail&lt;/a&gt;&lt;br&gt;&lt;iframe src="http://www.youtube.com/embed/Bw3ovKx5J5I" allowfullscreen&gt;&lt;/iframe&gt;</code>

**Figure 13. Multiple Multimedia Messages in the Information Column**

In the above example, you may notice there is a “&#39;” after “Cincinnati Children” text. The “&#39;” represents the apostrophe character in the html language. If your text message contains any apostrophe symbols, use “&#39;” to represent it.

If you are familiar with the website programming, like html, javascript, you can create more interesting styles for the multimedia message, like changing the fonts or colors of the texts, defining the size of the video player, etc. You can check reference [8] for more details.

## CREATE DIFFERENT LOCATION PIN ICONS

As seen in the above Google map report screenshots, the location pin icons on the Google map report can be different colors. Since each color of the icon is an image, simply put the website address of each icon in your icon column of the input data set to use that color icon.

Figure 14 is the screenshot for an icon column in the input data set that has 5 different colors for the location pin icons. Later, if you have other icons, you just copy the website address of that icon and put it to your input data set and you can call the macro in this paper to use it.

icon
<code>http://maps.google.com/mapfiles/ms/icons/blue-dot.png</code>
<code>http://maps.google.com/mapfiles/ms/icons/red-dot.png</code>
<code>http://maps.google.com/mapfiles/ms/icons/purple-dot.png</code>
<code>http://maps.google.com/mapfiles/ms/icons/yellow-dot.png</code>
<code>http://maps.google.com/mapfiles/ms/icons/green-dot.png</code>

**Figure 14. The Icon Column that has 5 Different Colors for the Location Pin Icons**

## THE MACRO CODE:

```
%macro generate_Google_map(indata=,reportnm=,lat=,lng=,tooltip=%str(),info=%str(),
icon=%str());
    *define the center location of the map;
    *using the mean value of latitude and longitude of the input data set;
    proc means data=&indata. noprint;
        var &lat. &lng.;
        output out=mapcenter(keep=map_center_lat map_center_lng)
            mean(&lat.)=map_center_lat mean(&lng.)=map_center_lng;
    run;

    %if "&icon." = "" %then %do;
        data tmp;
        set &indata.;
        icon="http://maps.Google.com/mapfiles/ms/icons/red-dot.png";
        run;
    %end;

    %else %if "&icon." ^= "" %then %do;
        data tmp;
        set &indata.;
        if &icon.=" then
            &icon.="http://maps.Google.com/mapfiles/ms/icons/red-dot.png";
        run;
    %end;

    *configure the Google map options;
    data _null_;
    file &reportnm.;
    set mapcenter;
    put '<html><head>'/
        '<script type="text/javascript"
        src="https://maps.googleapis.com/maps/api/js?v=3.exp&sensor=false"></script>'/
        '<script type="text/javascript">'/
        'function initialize() {' /
        'var map_center_point = new Google.maps.LatLng('map_center_lat','
        'map_center_lng');'/
        'var mapOptions = {' /
            'zoom: 4,' /
            'center: new Google.maps.LatLng(' map_center_lat ','
            map_center_lng '),'/
            'panControl: true,'/
            'zoomControl: true,'/
            'scaleControl: true,'/
            'mapTypeControl: true,'/
```

```

        'streetViewControl: true,'/
        'overviewMapControl: true,'/
        'mapTypeId: Google.maps.MapTypeId.TERRAIN};'/
'var map = new Google.maps.Map(document.getElementById("map-canvas"),
mapOptions);' /
'var infowindow = new Google.maps.InfoWindow();';
run;

*add the placed on to the map;
data _null_;
file &reportnm. mod lrecl=800;
set tmp;
i = compress(_n_);
%if "&info." ^="" %then %do;
    put 'var info' i '= "' '<div style="width:300px;height:300px;">
        <div style="float:left;margin-top:5px;margin-right:5px;"><b>' &info.
        "</b></div></div>" "';' /
%end;
%else %do;
    put
%end;
'var icon' i '= "' &icon. "';' /
%if "&tooltip." ^="" %then %do;
    'var tooltip' i '= "' &tooltip. "';' /
%end;
'var point' i '= new Google.maps.LatLng(' &lat. ', ' &lng. ');' /
'var marker' i '= new Google.maps.Marker({' /
    'position: point' i ', ' /
    'map: map,' /
    %if "&tooltip." ^="" %then %do;
        'title: tooltip' i ', '/
    %end;
    'icon: icon' i /
    '});' /

%if "&info." ^="" %then %do;
    'Google.maps.event.addListener(marker' i ', "click", function() {' /
        'infowindow.setContent(info' i ');' /
        'infowindow.open(map,marker' i ');' /
    '});';
%end;
%else %do;
    ;

```

```

%end;
run;

data _null_;
file &reportnm. mod;
put
' }'
'Google.maps.event.addDomListener(window, "load", initialize);' /
'</script>' /
'</head><body>' /
'<h3>Sample of Google map in SAS output with multiple points/markers</h3>' /
'<div id="map-canvas" style="width: 100%; height: 100%;"></div>' /
'</body></html>';
run;
proc sql;drop table tmp;quit;
%mend;

```

## CONCLUSION

The macro presented in this paper provides an easy way to generate an interactive Google map report. As long as you have an input data set that contains the longitude and latitude coordinates of the locations, you simply call the macro in this paper to generate a Google map report that has those locations on the map. Further, if you have the information for the tooltips, dialogue box, or the icons, you can prepare them in your input data set and call the macro to generate an interactive Google map report.

## REFERENCES

1. <http://support.sas.com/resources/papers/proceedings13/377-2013.pdf>
2. <http://www.mwsug.org/proceedings/2011/dataviz/MWSUG-2011-DG01.pdf>
3. <http://blogs.sas.com/content/sgf/2014/07/24/live-Google-maps-in-sas-multiple-markers/>
4. <http://support.sas.com/resources/papers/proceedings10/332-2010.pdf>
5. <http://support.sas.com/documentation/cdl/en/graphref/67288/HTML/default/viewer.htm#n0gc57iizcigign1luh4tssihzmy.htm>
6. [http://www.sascommunity.org/wiki/Geocoding\\_using\\_SAS\\_and\\_Google\\_Maps](http://www.sascommunity.org/wiki/Geocoding_using_SAS_and_Google_Maps)
7. <http://support.sas.com/resources/papers/proceedings09/015-2009.pdf>
8. <http://htmldog.com/reference/htmltags/>

## ACKNOWLEDGMENTS

The author wishes to thank the Division of Biostatistics and Epidemiology at Cincinnati Children's Hospital Medical Center for its support, and particularly Robert Tamer for his helpful feedback.

## CONTACT INFORMATION

Your comments and questions are valued and encouraged. Contact the author at:

Name: Ting Sa  
Enterprise: Cincinnati Children's Hospital Medical Center  
Address: 3333 Burnet Ave

City, State ZIP: Cincinnati, OH45229  
Work Phone: 513-636-3674  
E-mail: ting.sa@cchmc.org

SAS and all other SAS Institute Inc. product or service names are registered trademarks or trademarks of SAS Institute Inc. in the USA and other countries. © indicates USA registration.

Other brand and product names are trademarks of their respective companies.